[6.33] Sine and cosine integrals

The sine and cosine integrals are functions which sometimes come up in integration and engineering applications. For example, the sine integral shows up when evaluating the radiated power of antennas, and some hypergeometric functions The sine integral is called Si(z), and the cosine integral is called Ci(z). They are usually defined as

$$\mathrm{Si}(z)=\int_0^z \frac{\sin(t)}{t}dt \qquad \mathrm{Ci}(z)=\gamma+\ln(z)+\int_0^z \frac{\cos(t)-1}{t}dt$$

or sometimes defined as

$$\mathrm{Si}(z)=\frac{\pi}{2}-\int_z^\infty \frac{\sin(t)}{t}dt \qquad \mathrm{Ci}(z)=-\int_z^\infty \frac{\cos(t)}{t}dt$$

These symmetry relations hold:

$$\mathrm{Si}(-z)=-\mathrm{Si}(z) \qquad \mathrm{Ci}(-z)=\mathrm{Ci}(z)-i\pi \quad \text{where } 0<\arg(z)<\pi$$

where

$\gamma=0.577215664901533$ (Euler's constant)

Si(x) approaches $\pi/2$ as $x\to\infty$, and Ci(x) approaches 0 as $x\to\infty$. These series expansions also calculate the integrals:

$$\mathrm{Si}(z)=\sum_{n=0}^{\infty}\left[\frac{(-1)^n z^{2n+1}}{(2n+1)(2n+1)!}\right] \qquad \mathrm{Ci}(z)=\gamma+\ln(z)+\sum_{n=1}^{\infty}\left[\frac{(-1)^n z^{2n}}{2n(2n)!}\right]$$

This plot show the two functions from x = 0.1 to x = 15. The thin plot trace is Si(x), and the thick trace is Ci(x).



These functions can be calculated with the routines *Si()* and *Ci(),* shown below. These routines are not terribly accurate, but are good enough for most engineering applications. For arguments less than π , the built-in *nint()* function is used to integrate the function definition. For arguments greater than π , a rational approximation is used. The transition point of π was selected more for reasons of execution time than anything else. Above that point, *nint()* takes significantly longer to execute.

The routines must be installed in a folder called */spfn*. Set the modes to Approx, and Rectangular or Polar complex before execution. Input arguments must be real numbers, not complex numbers. Note that Ci(x) returns a complex number for arguments less than zero.

Because different methods are used for different argument ranges, there is a discontinuity in the function at $x=\pi$. The discontinuity is about 2.6E-7 for Si(x), and about 1.5E-8 for Ci(x). This may not practically affect the results when simply evaluating the function, but it may be a problem if you numerically differentiate the functions, or try to find a maximum or minimum with the functions.

The rational approximations used for Si(x) and Ci(x) are based on the identities

$$\mathrm{Si}(x) = \frac{\pi}{2} - f(x)\cos(x) - g(x)\sin(x)$$

$$\mathrm{Ci}(x) = f(x)\sin(x) - g(x)\cos(x)$$

where

$$f(x) = \frac{1}{x}\left(\frac{x^8+a_1x^6+a_2x^4+a_3x^2+a_4}{x^8+b_1x^6+b_2x^4+b_3x^2+b_4}\right)+\varepsilon(x) \qquad |\varepsilon(x)| < 5\cdot 10^{-7}$$

a_1 = 38.027264 b_1 = 40.021433
a_2 = 265.187033 b_2 = 322.624911
a_3 = 335.677320 b_3 = 570.236280
a_4 = 38.102495 b_4 = 157.105423

$$g(x) = \frac{1}{x^2}\left(\frac{x^8+a_1x^6+a_2x^4+a_3x^2+a_4}{x^8+b_1x^6+b_2x^4+b_3x^2+b_4}\right)+\varepsilon(x) \qquad |\varepsilon(x)| < 3\cdot 10^{-7}$$

a_1 = 42.242855 b_1 = 48.196927
a_2 = 302.757865 b_2 = 482.485984
a_3 = 352.018498 b_3 = 1114.978885
a_4 = 21.821899 b_4 = 449.690326

All of these equations are from *Handbook of Mathematical Functions*, Abramowitz and Stegun, Dover, 1965.

The results from both of these routines are only accurate to about seven significant digits. Below $x = \pi$, the accuracy may be as good as 11 or 12 significant digits. Notice that the maximum argument for both routines is 1E12 radians. This limit results from the use of the built-in *sin()* and *cos()* functions.

The evaluation of ci(x) and si(x) is also complicated by the fact that *nInt()* evaluates the integrals very slowly, and with poor accuracy, for arguments less than about 1E-4. To get around this, I used Taylor series expansions for the integrands near x=0, then symbolically integrated those. For the cosine integral, I use

$$\frac{\cos(t)-1}{t} = -\frac{t}{2}+\frac{t^3}{24}-\frac{t^5}{720}+\frac{t^7}{40320}+\cdots$$

and integrated this expansion to find

$$\mathrm{Ci}(x) \simeq -\frac{x^2}{4}+\frac{x^4}{96}-\frac{x^6}{4320}+\frac{x^8}{322560}$$

Similarly, for the sine integral:

$$\frac{\sin(t)}{t} = 1-\frac{t^2}{6}+\frac{t^4}{120}-\frac{t^6}{5040}+\frac{t^8}{362880}-\frac{t^{10}}{39916800}+\cdots$$

and integrating gives

$$\mathrm{Si}(x) \simeq x-\frac{x^3}{18}+\frac{x^5}{600}-\frac{x^7}{35280}+\frac{x^9}{3265920}-\frac{x^{11}}{439084800}$$

These approximations give good accuracy for x < 0.01.

Code listing for si(x):

```
si(x)
Func
©(x) return si(x), x real, |x|<1E12
©Must be in folder \spfn
©24jan01/dburkett@infinet.com

if x=0:return 0
if x<0:return -spfn\si(-x)

if x>π then

π/2-(((polyeval({1,0,38.027264,0,265.187033,0,335.677320,0,38.102495},x))/(polye
val({1,0,40.021433,0,322.624911,0,570.23628,0,157.105423},x)))/(x)*cos(x)+(polye
val({1,0,42.242855,0,302.757865,0,352.018498,0,21.821899},x)/polyeval({1,0,48.19
6927,0,482.485984,0,1114.978885,0,449.690326},x))/(x^2)*sin(x))

elseif x>.01 then

nint(sin(z)/z,z,0,x)

else

polyeval({-2.2774643986765E-9,0,3.0619243582207E-7,0,-2.8344671201814E-5,0,1.666
6666666667E-3,0,-.055555555555556,0,1,0},x)

endif

EndFunc
```

Code listing for ci(x):

```
ci(x)
Func
©(x) return ci(x), x real, |x|<1E12
©Must be in folder \spfn
©23jan01/dburkett@infinet.com

if x=0:return undef
if x<0:return spfn\ci(-x)+i*π

if x>π then

(((polyeval({1,0,38.027264,0,265.187033,0,335.677320,0,38.102495},x))/(polyeval(
{1,0,40.021433,0,322.624911,0,570.23628,0,157.105423},x)))/(x))*sin(x)-((polyeva
l({1,0,42.242855,0,302.757865,0,352.018498,0,21.821899},x)/polyeval({1,0,48.1969
27,0,482.485984,0,1114.978885,0,449.690326},x))/(x^2))*cos(x)

elseif x>.01 then

nint((cos(t)-1)/t,t,0,x)+ln(x)+.57721566490153

else

ln(x)+.57721566490153+polyeval({3.100198412698E-6,0,-2.3148148148148E-4,0,.01041
6666666667,0,-.25,0,0},x)

endif

EndFunc
```

(Thanks to Bhuvanesh for providing test data.)